

# Index notation in Lean 4

Joseph Tooby-Smith  
Reykjavik University

November 1, 2024

## Abstract

The physics community relies on index notation to effectively manipulate tensors of specific types. This paper introduces the first formally verified implementation of index notation in the interactive theorem prover Lean 4. By integrating index notation into Lean, we bridge the gap between traditional physics notation and formal verification tools, making it more accessible for physicists to write and prove results within Lean. In the background, our implementation leverages a novel application of category theory.

## 1. INTRODUCTION

**HepLean** In previous work, the author initiated the digitalization (or formalization) of high energy physics results using the interactive theorem prover Lean 4 in a project called HepLean. Lean is an interactive theorem prover and programming language with syntax resembling traditional pen-and-paper mathematics. Lean allows users to write definitions, theorems, and proofs which are then automatically checked for correctness, using its foundation of dependent-type theory. HepLean has four main motivations: A linear storage of information makes look-up of results easier; Allows for the creation and proof of new results using automated tactics like AI; Makes it easier to check the correctness of results; And allows for new ways high-energy physics and computer science can be taught.

**Other works in Lean** HepLean is part of a broader movement of projects to formalize parts, or all, of mathematics and science. The largest of these projects is Mathlib, which aims to formalize mathematics. Indeed, HepLean is built downstream of Mathlib, meaning it has Mathlib as a dependency and uses many of the definitions and theorems from there. Other projects include the ongoing effort led by Kevin Buzzard to formalize the proof of Fermat's Last Theorem into Lean. In the realm of the sciences, `js:`

sorry

**Notation in physics and index notation:** Physicists heavily rely on specialized notation to express complex concepts succinctly. Among these, index notation is particularly prevalent, as it provides a compact and readable way to represent specific types of tensors and operations between them.

**Challenge:** Formalizing index notation into an interactive theorem prover like Lean is challenging due to the complexity and implicitness of the notation and the need for flexible and rigorous formalization. This paper presents such an implementation of index notation within the HepLean project.

**Motivation:** The primary motivations for implementing index notation into Lean are:

1. To make easier to write and prove results from high energy physics into Lean.
2. To make the syntax of Lean more familiar to high energy physicists use to index notation.

We hope that the implementation of index notation into Lean not only enhances usability but also promotes the adoption of formal methods in the physics community.

**Conclusion:** The conclusion of the implementation of index notation into Lean 4 is that we can write results like the following:

$$\{\text{pauliCo} \mid \forall \alpha \beta \otimes \text{pauliContr} \mid \forall \alpha' \beta' = 2 \cdot \epsilon_L \mid \alpha \alpha' \otimes \epsilon_R \mid \beta \beta'\}^T$$

and lean will correctly interpret this result as a formal mathematical object which proves can be applied to. [js: ref](#) Mention that we will return to this example later.

**Other papers:** Previous attempts to formalize index notation have been made in programming languages like Haskell [?]. However, these implementations do not provide the formal verification capabilities inherent in Lean. The formal verification requirement of Lean introduces unique challenges in implementing index notation, necessitating (what we believe is) a novel solution.

**Outline:** This paper is split into two main sections. Section [js: ref](#) discusses the implementation of index notation into Lean. Section [js: ref](#) gives two examples of theorems and proofs using index notation. A more minor section, section [js: ref](#) discusses the future work related to this project.

## 2. IMPLEMENTATION OF INDEX NOTATION INTO LEAN 4



The implementation of index notation in Lean can be broken down into three main components, illustrated in Figure [js: ref](#). The first component is the *syntax for tensor expressions*, which is what users interact with when writing results in Lean. This syntax closely mirrors the notation familiar to physicists, making it intuitive and accessible. It appears directly in the Lean files and can be thought of as an informal string that represents the tensor expressions.

The second component involves transforming this syntax into a *tensor tree*. The tensor tree is a formal mathematical representation of the tensor expression. By parsing the informal syntax into a structured tree, we establish a rigorous foundation that captures the tensor expression. This formal representation allows us to easily manipulate tensor expressions and prove results related to them in a way that Lean accepts as formal.

The third and final component is the conversion of the tensor tree into an actual *tensor*. This process utilizes properties of the symmetric-monoidal category of representations to translate the abstract tensor tree into a concrete tensor.

These three steps—syntax, tensor tree, and tensor—are illustrated in Figure [js: ref](#). Although Lean processes information from left to right, starting with the syntax and proceeding to the tensor tree, and then finally (when we ask it to) to the tensors, it is more effective to discuss the implementation from right to left. This reverse approach is advantageous because the tensors, which are the primary objects of interest, are located on the right side of the diagram. The left and middle parts of the diagram represent intermediate stages that facilitate the manipulation and understanding of these tensors and their expressions.

## 2.1. DEFINING TENSORS

### 2.1.1 Building blocks of tensors and color

Tensors of a species, such as complex Lorentz tensors, are constructed from a set of building block representations of a group  $G$  over a field  $k$ . For complex Lorentz tensors, the group  $G$  is  $SL(2, \mathbb{C})$ , the field  $k$  is the field of complex numbers.

There are six building block representations for complex Lorentz tensors. These are

- `Fermion.leftHanded` is the representation of  $SL(2, \mathbb{C})$  corresponding  $v \mapsto Mv$ , corresponding to Left-handed Weyl fermions.
- `Fermion.altLeftHanded` is the representation of  $SL(2, \mathbb{C})$  corresponding  $v \mapsto M^{-1T}v$ , corresponding to alternative Left-handed Weyl fermions (as we will call them).
- `Fermion.rightHanded` is the representation of  $SL(2, \mathbb{C})$  corresponding  $v \mapsto M^*v$ , corresponding to Right-handed Weyl fermions.
- `Fermion.altRightHanded` is the representation of  $SL(2, \mathbb{C})$  corresponding  $v \mapsto M^{-1\dagger}v$ , corresponding to alternative Right-handed Weyl fermions.
- `Lorentz.complexContr` is the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the contravariant action of the Lorentz group on four-vectors
- `Lorentz.complexCo` is the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the covariant action of the Lorentz group on four-vectors.

As an example the representation `Fermion.leftHanded` is defined in Lean as follows:

```
def leftHanded : Rep ℂ SL(2,ℂ) := Rep.of {
  toFun := fun M => {
    toFun := fun (ψ : LeftHandedModule) =>
      LeftHandedModule.toFin2CEquiv.symm (M.1 * ψ.toFin2C),
    map_add' := by
      intro ψ ψ'
      simp [mulVec_add]
    map_smul' := by
      intro r ψ
      simp [mulVec_smul]}
  map_one' := by
    ext i
    simp
  map_mul' := fun M N => by
    simp only [SpecialLinearGroup.coe_mul]
  ext1 x
    simp only [LinearMap.coe_mk, AddHom.coe_mk, LinearMap.mul_apply,
      LinearEquiv.apply_symm_apply,
      mulVec_mulVec]}
```

js: Not exact definiton as could not type `*_v` Here the outer `toFun` argument is defining a map from  $SL(2, \mathbb{C})$  to linear maps from `LeftHandedModule` (equivalent to  $\mathbb{C}^2$ ) to itself. The inner `toFun`, `map_add'`

and `map_smul` are defining respectively, the underlying function of this linear map, and proving it is linear with respect to addition and scalar multiplication. The `map_one` and `map_mul` are proving that action of the identity is trivial, and that the action of the product of two elements is the product of the actions of the elements.

We assign a unique label, which we refer to as a *color*, to each of the building block representations. We denote the type of colors for a given species of tensors as  $C$ . For complex Lorentz tensors  $C$  is defined as

```
inductive Color
| upL : Color
| downL : Color
| upR : Color
| downR : Color
| up : Color
| down : Color
```

Here `Color` is the name of our type and `upL`, `downL`, etc. are the colors.

To formally associate each color with its corresponding representation, we define a discrete functor from the set  $C$  to the category of  $k$ -representations of  $G$ , denoted  $\text{Rep}kG$ , that is, a functor

$$F_D : C \Rightarrow \text{Rep}kG. \quad (1)$$

For complex Lorentz tensors this functor in Lean as follows:

```
FDiscrete := Discrete.functor fun c =>
  match c with
  | Color.upL => Fermion.leftHanded
  | Color.downL => Fermion.altLeftHanded
  | Color.upR => Fermion.rightHanded
  | Color.downR => Fermion.altRightHanded
  | Color.up => Lorentz.complexContr
  | Color.down => Lorentz.complexCo
```

The reason why we have used a functor here, rather than just a function, will become apparent in what follows.

### 2.1.2 A general tensor

For a given tensor species, given the symmetric monoidal structure on  $\text{Rep}kG$ , we can take the tensor product of the building block representations. Elements of such tensor products form the general notion of a tensor for that given species.

To formalize this in Lean, we consider the category  $\mathcal{S}_{/C}^\times$ . The objects of  $\mathcal{S}_{/C}^\times$  are functions  $f : X \rightarrow C$  for some type  $X$ . A morphism from  $f : X \rightarrow C$  to  $g : Y \rightarrow C$  is a bijection  $\phi : X \rightarrow Y$  such that  $f = g \circ \phi$ . This category is equivalent to the core of the category of types ( $f$ ) over  $C$ , hence our notation.

Any functor  $H : C \Rightarrow \text{Rep}kG$  can be lifted to a symmetric monoidal functor  $\mathcal{S}_{/C}^\times \Rightarrow \text{Rep}kG$  which takes  $f$  to the tensor product  $\bigotimes_{x \in X} H(f(x))$  and morphisms to the linear maps of representations corresponding to reindexings of tensor products.

This construction is itself functorial, giving a functor:

$$\text{lift} : \text{Fun}(C, \text{Rep}kG) \Rightarrow \text{SymmMonFun}(\mathcal{S}_C^\times, \text{Rep}kG) \quad (2)$$

In the previous subsection we defined the functor  $F_D$  which associates to each color its corresponding representation. We can then define a functor  $F$  by the image of  $F_D$  under the lift functor.

A tensor of a given species can then be thought of as a vector in the representations in the of  $F$ . Defining a tensor in this way allows us to utalize the structur of monodial categories and functors in a useful way.

In HepLean we formally define lift and  $F$ .

For most purposes in physics  $X$  will be a type  $\text{Finn}$ , corresponding to the type (set) of natural numbers (including 0) less  $n$ . Later on we will restrict to these types.

### 2.1.3 Basic operations

Now we have discussed how tensors of a given species can be formally defined, we can define basic operations on these tensors.

The simipalist of these operations are addition, scalar multiplication and group action. Addition and scalar multiplication are given to us for free from the vector space structure on  $F(f)$ . The action of the group  $G$  on tensors also comes for free from the fact that  $F(f)$  lives in  $\text{Rep}kG$ .

The next simplest operation is permutation of indices (or building block representations). Such permutations arise from applying  $F$  to morphisms in  $\mathcal{S}_C^\times$ . Theses sorts of permutations are often implicit in the notation physicists use, but arise in tensor expressions such as  $T_{\mu\nu} = T_{\nu\mu}$ .

The next operation is the tensor product of two tensors. Given  $f : X \rightarrow C$  and  $g : Y \rightarrow C$ , and tensors  $t \in F(f)$  and  $s \in F(g)$  we can form a vector in  $F(f) \otimes F(g)$ . Formally this can be done by taking the tensor product of the morphisms  $\mathbb{K} \rightarrow F(f)$  and  $\mathbb{K} \rightarrow F(g)$  in the category of vector spaces over  $k$ . We can then use the tensorate of  $F$  to get a tensor in  $F(f \otimes g)$ .

We now turn to the slightly more complicated operation of contraction of indices, and the related metric and unit. To define the contraction of introduce an involution  $\tau : C \rightarrow C$ . For complex Lorentz tensors this is defined as follows

```
τ := fun c =>
  match c with
  | Color.upL => Color.downL
  | Color.downL => Color.upL
  | Color.upR => Color.downR
  | Color.downR => Color.upR
  | Color.up => Color.down
  | Color.down => Color.up
```

We say that  $\tau$  takes a color to it's dual. With  $\tau$  and  $F_D$  we can define the functor  $F_\tau : C \Rightarrow \text{Rep}kG$  which takes  $c$  to  $F(c) \otimes F(\tau c)$ . Letting  $\mathbb{K}$  be the constant functor from  $C$  landing on the, we basic data for contraction is contained in a natural transformation from  $F_\tau$  to  $\mathbb{K}$ . That is, for each  $c \in C$  an equivariant map from  $F(c) \otimes F(\tau c)$  to the trivial representation.

To see how this is can be used to contract indices, consider a tensor  $t \in F(f)$ , for  $f : \text{Fin } n.\text{succ}.\text{succ} \rightarrow C$ . Here  $\text{Fin } n.\text{succ}.\text{succ}$  is the type of all number  $0, 1, 2, \dots, n+1$  less than  $n.\text{succ}.\text{succ} = n+2$ . Choosing

two distinct indices  $i, j : \text{Fin } n.\text{succ}.\text{succ}$  we can contract them using the following chain of maps

$$Ff \equiv F_\tau(fi) \otimes F(f'i) \rightarrow \mathbb{K} \otimes F(f'i) \rightarrow F(f'i) \quad (3)$$

Here  $f' : \text{Fin } n \rightarrow C$  is the function derived from  $f$  by removing the indices  $i$  and  $j$ . The first equivalence is somewhat complicated to define formally, but essentially involves extracting  $i$  and  $j$ .

The metric and unit are defined in a similar way. The metric is a natural transformation from  $\mathbb{K}$  to  $F_D \otimes F_D$ . That is, for each  $c \in C$  an equivariant map from the trivial representation to  $F_D(c) \otimes F_D(c)$ . The unit is a natural transformation from  $\mathbb{K}$  to  $F'_\tau$  which takes  $c$  to  $F_D(\tau c) \otimes F_D(c)$ . That is, for each  $c \in C$  an equivariant map from the trivial representation to  $F_D(\tau c) \otimes F_D(c)$ .

The choice of contraction, metric and unit are subject to a number of conditions.

- Contraction must be symmetric.

$$\begin{array}{ccc} F_D(c) \otimes F(\tau c) & \longrightarrow & F_D(\tau c) \otimes F_D(c) \\ \downarrow & & \downarrow \\ \mathbb{I} & \longrightarrow & F_D(\tau c) \otimes F_D(\tau \tau c) \end{array} \quad (4)$$

- Contraction with the unit does nothing.

$$\begin{array}{ccccc} F_D(c) & \longrightarrow & F_D(c) \otimes \mathbb{I} & \longrightarrow & F_D(c) \otimes (F_D(\tau c) \otimes F_D(c)) \\ \uparrow & & & & \downarrow \\ \mathbb{I} \otimes F_D(c) & \longleftarrow & & & (F_D(c) \otimes F_D(\tau c)) \otimes F_D(c) \end{array} \quad (5)$$

- The unit is symmetric

$$\begin{array}{ccc} \mathbb{I} & \longrightarrow & F_D(\tau c) \otimes F_D(c) \\ \downarrow & & \uparrow \\ F_D(\tau \tau c) \otimes F_D(\tau c) & \longrightarrow & F_D(\tau c) \otimes F_D(\tau \tau c) \end{array} \quad (6)$$

- Contraction with the metric its dual returns the unit

$$\begin{array}{ccc} \mathbb{I} & \longrightarrow & F_D(\tau c) \otimes F_D(c) \\ \downarrow & & \uparrow \\ (F_D c \otimes F_D c) \otimes (F_D(\tau c) \otimes F_D(\tau c)) & & F_D(c) \otimes F_D(\tau c) \\ \downarrow & & \uparrow \\ F_D c \otimes (F_D c \otimes (F_D(\tau c) \otimes F_D(\tau c))) & \longrightarrow & F_D c \otimes ((F_D c \otimes F_D(\tau c)) \otimes F_D(\tau c)) \end{array} \quad (7)$$

In Lean we quite write these conditions in this way, instead we give them in a simpler equivalent form in which we apply the maps in the above diagram to pure tensors. As an example, the condition is written in Lean as

The last operation we want to talk about is evaluation. Where one specifies the exact value of one of the indices of a tensor e.g.  $\eta_{0i}$ . All of the operations discussed thus far have related to the category  $\text{Rep} kG$ . As such, they have respect the group action. Evaluation, on the other hand, does not respect the group

action. For example, taking the 0th component of a four-vector is not Lorentz invariant. Nevertheless, we can define evaluation in the category of vector spaces. To do this, we need a basis for each  $F_D(c)$ , the coordinates of a vector in  $F_D(c)$  in that basis is described by a linear map of vector spaces from  $F_D(c)$  to  $k$ . This can be used to evaluate the index of a tensor in the following way:

$$Ff \equiv F_D(fi) \otimes F(f'i) \rightarrow k \otimes F(f'i) \rightarrow F(f'i) \quad (8)$$

where  $f'$  is  $f$  with the  $i$ th index removed, and again the first equivalence is somewhat complicated to define but involves extracting the  $i$ th index.

#### 2.1.4 Tensor Species

The data of the field  $k$ , the group  $G$ , the functor  $F_D$  (from which  $F$  can be derived), the involution  $\tau$ , the natural transformations for contraction, metric, and unit, and basis needed for evaluation, form formally what we have loosely been calling a Tensor species.

That is, the difference between complex Lorentz tensors, Einstein tensors, and real Lorentz tensors is down to this data. In Lean it is useful to work with general tensor species where possible, so important results have to be defined only once. Thus we make the following definition

```

/-- The sturcture of a type of tensors e.g. Lorentz tensors, Einstien tensors,
    complex Lorentz tensors. -/
structure TensorSpecies where
  /-- The colors of indices e.g. up or down. -/
  C : Type
  /-- The symmetry group acting on these tensor e.g. the Lorentz group or  $SL(2, \mathbb{C})$ .
      -/
  G : Type
  /-- An instance of 'G' as a group. -/
  G_group : Group G
  /-- The field over which we want to consider the tensors to live in, usually ' $\mathbb{R}$ '
      or ' $\mathbb{C}$ '. -/
  k : Type
  /-- An instance of 'k' as a commutative ring. -/
  k_commRing : CommRing k
  /-- A 'MonoidalFunctor' from 'OverColor C' giving the rep corresponding to a map
      of colors
      'X  $\rightarrow$  C'. -/
  FDiscrete : Discrete C  $\Rightarrow$  Rep k G
  /-- A map from 'C' to 'C'. An involution. -/
   $\tau$  : C  $\rightarrow$  C
  /-- The condition that ' $\tau$ ' is an involution. -/
   $\tau$ _involution : Function.Involutive  $\tau$ 
  /-- The natural transformation describing contraction. -/
  contr : OverColor.Discrete.pair  $\tau$  FDiscrete  $\tau \rightarrow \mathbb{1}_$  (Discrete C  $\Rightarrow$  Rep k G)
  /-- The natural transformation describing the metric. -/
  metric :  $\mathbb{1}_$  (Discrete C  $\Rightarrow$  Rep k G)  $\rightarrow$  OverColor.Discrete.pair FDiscrete
  /-- The natural transformation describing the unit. -/
  unit :  $\mathbb{1}_$  (Discrete C  $\Rightarrow$  Rep k G)  $\rightarrow$  OverColor.Discrete. $\tau$ Pair FDiscrete  $\tau$ 
  /-- A specification of the dimension of each color in C. This will be used for
      explicit
      evaluation of tensors. -/
  repDim : C  $\rightarrow$   $\mathbb{N}$ 
  /-- repDim is not zero for any color. This allows casting of ' $\mathbb{N}$ ' to 'Fin
      (S.repDim c)'. -/
  repDim_neZero (c : C) : NeZero (repDim c)
  /-- A basis for each Module, determined by the evaluation map. -/
  basis : (c : C)  $\rightarrow$  Basis (Fin (repDim c)) k (FDiscrete.obj (Discrete.mk c)).V
  /-- Contraction is symmetric with respect to duals. -/
  contr_tmul_symm (c : C) (x : FDiscrete.obj (Discrete.mk c))
    (y : FDiscrete.obj (Discrete.mk ( $\tau$  c))) :
    (contr.app (Discrete.mk c)).hom (x  $\otimes_{\tau[k]}$  y) = (contr.app (Discrete.mk ( $\tau$ 
    c))).hom
    (y  $\otimes_{\tau}$  (FDiscrete.map (Discrete.eqToHom ( $\tau$ _involution c).symm)).hom x)
  /-- Contraction with unit leaves invariant. -/
  contr_unit (c : C) (x : FDiscrete.obj (Discrete.mk (c))) :
    ( $\lambda_$  (FDiscrete.obj (Discrete.mk (c))))).hom.hom
    (((contr.app (Discrete.mk c))  $\triangleright$  (FDiscrete.obj (Discrete.mk (c))))).hom
    (( $\alpha_$  _ _ (FDiscrete.obj (Discrete.mk (c))))).inv.hom
    (x  $\otimes_{\tau[k]}$  (unit.app (Discrete.mk c)).hom (1 : k)))) = x
  /-- The unit is symmetric. -/
  unit_symm (c : C) :
    ((unit.app (Discrete.mk c)).hom (1 : k)) =

```



We can then let e.g.  $S : \text{TensorSpecies}$  and recover the functor  $F_D$  by  $S.F\text{Discrete}$ . The functor  $F$  is defined separately can be recovered by  $S.F$ . If  $f : X \rightarrow S.C$  is a map then a tensor is an element of  $S.F.\text{obj}$  ( $\text{OverColor.mk } f$ ). The  $\text{OverColor.mk}$  tells Lean to consider  $f$  as a function an element of the category  $\mathcal{S}_{/C}^X$ , or as it is written in Lean  $\text{OverColor } S.C$ .

A function  $f : \text{Fin } 2 \rightarrow S.C$ , for instance landing on  $c1, c2 \in C$  can be written in Lean as  $![c1, c2]$ , thus we can write  $S.F.\text{obj} (\text{OverColor.mk } ![c1, c2])$ .

We will see instances of this for Complex Lorentz tensors. As an example, a Lorentz tensor  $T_\nu^\mu$  can be written as

```
T : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.down])
```

## 2.2. TENSOR TREES

A tensor expression consists of a series of tensors and operations between them. For example

$$\eta_{\mu\nu} P_\sigma^\nu + V_{\mu\sigma}, \quad (9)$$

consists of a product of tensors, a contraction and a addition. Such an expression is a tensor in its own right, and we could just use the operations discussed above to define it.

However, it is useful for bridging the gap between syntax and a tensor to have a more structured way of representing such expressions. This is where tensor trees come in. Tensor trees will also be useful when it comes to proving results about tensors.

A tensor tree is essentially a tree with nodes representing tensors or operations on tensors. For example the tensor tree for the expression above is:

Since we really only care about tensors with  $X = \text{Finn}$ , tensor trees in Lean are implemented only for these.

In Lean we define a tensor tree as follows:

```

/-- A syntax tree for tensor expressions. -/
inductive TensorTree (S : TensorSpecies) : {n : ℕ} → (Fin n → S.C) → Type where
  /-- A general tensor node. -/
  | tensorNode {n : ℕ} {c : Fin n → S.C} (T : S.F.obj (OverColor.mk c)) :
    TensorTree S c
  /-- A node corresponding to the addition of two tensors. -/
  | add {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c → TensorTree
    S c
  /-- A node corresponding to the permutation of indices of a tensor. -/
  | perm {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
    (σ : (OverColor.mk c) → (OverColor.mk c1)) (t : TensorTree S c) :
    TensorTree S c1
  /-- A node corresponding to the product of two tensors. -/
  | prod {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
    (t : TensorTree S c) (t1 : TensorTree S c1) : TensorTree S (Sum.elim c c1 o
    finSumFinEquiv.symm)
  /-- A node corresponding to the scalar multiple of a tensor by a element of the
    field. -/
  | smul {n : ℕ} {c : Fin n → S.C} : S.k → TensorTree S c → TensorTree S c
  /-- A node corresponding to negation of a tensor. -/
  | neg {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c
  /-- A node corresponding to the contraction of indices of a tensor. -/
  | contr {n : ℕ} {c : Fin n.succ.succ → S.C} : (i : Fin n.succ.succ) →
    (j : Fin n.succ) → (h : c (i.succAbove j) = S.τ (c i)) → TensorTree S c →
    TensorTree S (c o Fin.succAbove i o Fin.succAbove j)
  /-- A node corresponding to the action of a group element on a tensor. -/
  | action {n : ℕ} {c : Fin n → S.C} : S.G → TensorTree S c → TensorTree S c
  /-- A node corresponding to the evaluation of an index of a tensor. -/
  | eval {n : ℕ} {c : Fin n.succ → S.C} : (i : Fin n.succ) → (x : ℕ) →
    TensorTree S c →
    TensorTree S (c o Fin.succAbove i)

```

Let us give an example of what this notation means. For each  $n : \mathbb{N}$  and map  $c : \text{Fin } n \rightarrow S.C$  we get type  $\text{TensorTree } S \ c$ . This is the type of tensor trees corresponding to tensors in  $S.F.\text{obj } (\text{OverColor.mk } c)$ .

The first constructor `tensorNode` generates a tensor tree of type  $\text{TensorTree } S \ c$  from a tensor of type  $S.F.\text{obj } (\text{OverColor.mk } c)$ . The other constructors are slightly more complicated. The constructor `contr` says given an index  $i$  and  $j$ , and a tensor tree  $t$  of type  $\text{TensorTree } S \ c$ , we can get a tensor tree of type  $\text{TensorTree } S \ (c \circ \text{Fin.succAbove } i \circ \text{Fin.succAbove } j)$ .

So in Lean the expression above could be written as follows: `js: ...`

This definition of a tensor tree does not rely the actual nature of the operations invovled. From a tensor tree we can define a tensor itself using a recursively defined map

```

/-- The underlying tensor a tensor tree corresponds to. -/
def tensor : ∀ {n : ℕ} {c : Fin n → S.C}, TensorTree S c → S.F.obj (OverColor.mk
  c) := fun
| tensorNode t => t
| add t1 t2 => t1.tensor + t2.tensor
| perm σ t => (S.F.map σ).hom t.tensor
| neg t => - t.tensor
| smul a t => a · t.tensor
| prod t1 t2 => (S.F.map (OverColor.equivToIso finSumFinEquiv).hom).hom
  ((S.F.μ _ _).hom (t1.tensor ⊗ t2.tensor))
| contr i j h t => (S.contrMap _ i j h).hom t.tensor
| eval i e t => (S.evalMap i (Fin.ofNat' e Fin.size_pos')) t.tensor
| action g t => (S.F.obj (OverColor.mk _)).ρ g t.tensor

```

Note that this definition is recursive, Lean can automatically determine that this will terminate (with a little help about the number of nodes in a tensor tree). This association of a tensor tree with a tensor is not one-to-one (but it is onto). Many tensor trees can represent the same tensor, in the same way that many tensor expressions represent the same tensor.

### 2.2.1 Using Tensor trees in proofs

Tensor trees can be used in proofs for the following reason. Define a sub-tree of a tensor trees to be a node and all child nodes of that node. If  $T$  is a tensor tree and  $S$  a sub-tree of  $T$ , we can replace  $S$  in  $T$  with another tensor tree  $S'$  to get a new overall tensor-tree  $T'$ . If  $S$  and  $S'$  have the same underlying tensor, then  $T$  and  $T'$  will.

In Lean this property is encoded in a number of lemmas, for example: [js: lemma](#)

We will this at play in the example section of this paper.

## 2.3. ELABORATION

We now dicuss how we make the Lean code look similar to what we would use on pen-and-paper physics.

This is done using a two step process. Firstly, we define syntax for tensor expressions. Then we write code to turn this syntax into a tensor tree. This process is not formally defined or verified, Lean takes the outputted tensor tree as the formal object to work with.

Instead of deleving into the nitty-gritty details of how this process works under the hood, we give some examples to see how it works.

In what follows we will assume that  $T$ , and  $T_i$  etc are tensors defined as  $S.F.obj\_$  for some tensor species  $S$ . We will also assume that these tensors are defined correctly for the expressions below to make sense.

The syntax allows us to write the following

$\{T \mid \mu \ v\}^T$	tensorNode T
------------------------	--------------

for a tensor node. Here the  $\mu$  and  $v$  are free variables and it does not matter what we call them - Lean will elaborate the expression in the same way. The elaborator also knows how many indices to expect

for a tensor  $T$  and will raise an error if the wrong number are given. The  $\{\_ \}^T$  notation is used to tell Lean that the syntax is to be treated as a tensor expression.

We can write e.g.

$\{T \mid \mu \ v\}^T$	<code>.tensor (tensorNode T).tensor</code>
------------------------	--

to get the underlying tensor.

Note that we have not lowered or risen the indices, as one would expect from pen-and-paper notation. There is one primary reason for this, whether an index is lower or risen does not carry any information, since this information comes from the tensor itself. Also, for something like complex Lorentz tensors, there are at least three different types of upper-index.

We can extract the tensor from a tensor tree using the following syntax. If we want to evaluate an index we can put an explicit index in place of  $\mu$  or  $v$  above, for example

$\{T \mid 1 \ v\}^T$	<code>eval 0 1 (tensorNode T)</code>
----------------------	--------------------------------------

The syntax and elaboration for negation, scalar multiplication and the group action are fairly similar. For negation we have

$\{T \mid \mu \ v\}^T$	<code>neg (tensorNode T)</code>
------------------------	---------------------------------

For scalar multiplication by  $a \in k$  we have

$\{a \cdot T \mid \mu \ v\}^T$	<code>smul a (tensorNode T)</code>
--------------------------------	------------------------------------

For the group action of  $g \in G$  on a tensor  $T$  we have

$\{g \cdot_a T \mid \mu \ v\}^T$	<code>action g (tensorNode T)</code>
----------------------------------	--------------------------------------

For the product of two tensors is also fairly simple, we have

$\{T \mid \mu \ v \otimes T2 \mid \sigma\}^T$	<code>prod (tensorNode T) (tensorNode T2)</code>
---	--

The syntax for contraction is as one expect,

$\{T \mid \mu \ v \otimes T2 \mid v \ \sigma\}^T$	<code>contr 1 1 rfl (prod (tensorNode T) (tensorNode T2))</code>
---	--

On the RHS here the first argument (1) of `contr` is the index of the first  $v$  on the LHS, the second argument (also 1) is the second index. The `rfl` is a proof that the colors of the two contracted indices are actually dual to one another. If they are not, this proof will fail and the elaborator will complain. It will also complain if more then two indices are trying to be contracted. Although, this depends on where exactly the indices sit in the expression. For example

$\{T \mid \mu \ v \otimes T2 \mid v \ v\}^T$	<code>(prod (tensorNode T) (contr 0 0 rfl (tensorNode T2)))</code>
--	--

works fine because the inner contraction is done before the product.

We now turn to addition. Our syntax allows for e.g.  $\{T \mid \mu \ v + T2 \mid \mu \ v\}^T$  and also  $\{T \mid \mu \ v + T2 \mid v \ \mu\}$ , provided of course that the indices are of the correct color (which Lean will check). The elaborator handles both these cases, and generalizations thereof by adding a permutation node. Thus we have

$\{T \mid \mu \ v + T2 \mid \mu \ v\}^T$	<code>add (tensorNode T) (perm _ (tensorNode T2))</code>
--	--

where here the  $\_$  is a placeholder for the permutation, and in this case will be trivial, but for

$\{T \mid \mu \nu + T2 \mid \nu \mu\}^T$	<code>add (tensorNode T) (perm _ (tensorNode T2))</code>
--	--

it will be the permutation for the two identities.

Despite not forming part of a node in our tensor tree, we also give syntax for equality. This is done in a very similar way to addition, with the addition of a permutation node to account for e.g. the fact that  $T_{\mu\nu} = T_{\nu\mu}$ .

$\{T \mid \mu \nu = T2 \mid \nu \mu\}^T$	<code>(tensorNode T).tensor = (perm _ (tensorNode T2)).tensor</code>
--	--

Note the use of the `.tensor` to extract the tensor from the tensor tree, it does not really mean much to ask for equality of the tensor trees themselves.

### 3. EXAMPLES

We give two examples in this section. The first example is a simple theorem involving index notation and tensor trees. We will show here, in rather explicit detail, how we can manipulate tensor trees to solve such theorems. The second example we shall give will show a number of definitions in HepLean concerning index notation. Here we will not give so much detail, the point rather being to show the reader the broad use of our construction.

#### 3.1. EXAMPLE 1: SYMMETRIC AND ANTI-SYMMETRIC TENSOR

If  $A^{\mu\nu}$  is an anti-symmetric tensor and  $S_{\mu\nu}$  and  $S$  is a symmetric tensor, then it is true that  $A^{\mu\nu}S_{\mu\nu} = -A^{\mu\nu}S_{\mu\nu}$ . In Lean this result, and it's lemma are written as follows:

```
lemma antiSymm_contr_symm
  {A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
  {S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
  (hA : {A |  $\mu \nu = - (A | \nu \mu)\}^T$ ) (hs : {S |  $\mu \nu = S | \nu \mu\}^T$ ) :
  {A |  $\mu \nu \otimes S | \mu \nu = - A | \mu \nu \otimes S | \mu \nu\}^T := by
conv =>
  lhs
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_fst <| hA]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_snd <| hs]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_perm_left _ _ _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_tensor_eq <| prod_perm_right _ _
    _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_perm _ _]
  rw [contr_tensor_eq <| perm_contr_congr 1 2]
  rw [perm_contr_congr 0 0]
  rw [perm_tensor_eq <| contr_contr _ _]
  rw [perm_perm]
  rw [perm_tensor_eq <| contr_tensor_eq <| contr_tensor_eq <| neg_fst_prod _ _]
  rw [perm_tensor_eq <| contr_tensor_eq <| neg_contr _]
  rw [perm_tensor_eq <| neg_contr _]
  apply perm_congr _ rfl
  decide$ 
```

Let us break this down. The statements

```
{A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
{S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
```

are simply defining  $A$  and  $S$  to be tensors of type  $A^{\mu\nu}$  and  $S_{\mu\nu}$  respectively. This agrees with the notation set out in §?? [js: ref](#).


The parameter `hA` is stating that  $A$  is anti-symmetric. Expanded in terms of tree diagrams we have

$hA : \{A \mid \mu \nu = - (A \mid \nu \mu)\}^T$	
Description: The tensor $A$ is anti-symmetric.	

Similarly, the parameter `hS` is stating that  $S$  is symmetric. Expanded in terms of tree diagrams

$hS : \{S \mid \mu \nu = S \mid \nu \mu\}^T$	
Description: The tensor $S$ is symmetric.	

The line  $\{A \mid \mu \nu \otimes S \mid \mu \nu = - A \mid \mu \nu \otimes S \mid \mu \nu\}^T$  is the statment we are trying to prove. In terms of tree diagrams it says that




The `perm` here actually does nothing, but is included by Lean.

The lines of the proof in the `conv` block are manipulations of the tensor tree on the LHS of the equation. The `rw` tactic is used to rewrite the tensor tree using the various lemmas. We go through each step in turn.

```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_fst <| hA]
```

---


Description: Replace the node A with the RHS of hA.



```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_snd <| hS]
```

---


Description: Replace the node S with the RHS of hS.



```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_perm_left _ _ _]
```


---

Description: Move the left permutation through the product.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| perm_tensor_eq <| prod_perm_right _ _
_ _]
```

Description: Move the right permutation through the product.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| perm_perm _ _ _]
```

Description: Combine the two permutations (using functoriality).



```
rw [contr_tensor_eq <| perm_contr_congr
1 2]
```


Description: Move the permutation through the contraction. And simplify the contraction indices to 1 and 2 (Lean will check if this is correct).






```
rw [perm_contr_congr 0 0]
```

Description: Move the permutation through the contraction. And simplify the contraction indices to 0 and 0 (Lean will check if this is correct).




```
rw [perm_tensor_eq <| contr_contr _ _ _]
```

Description: Swap the two contractions. This introduces a permutation.




```
rw [perm_perm]
```

Description: Combine the two permutations.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  contr_tensor_eq <| negfst_prod _ _]
```

Description: Move the negation through the product.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  neg_contr _]
```

Description: Move the negation through the first contraction.



```
rw [perm_tensor_eq <| neg_contr _]
```

Description: Move the negation through the second contraction.



The remainder of the proof `apply perm_congr _ rfl` and decide that the tensor trees on the LHS and RHS are actually equal.

### 3.2. EXAMPLE 2: PAULI MATRICES AND BISPINORS

Using the formism we have set up thus far it is possible to define Pauli matrices and bispinors as complex Lorentz tensors.

The pauli matrices appear in HepLean as follows

```

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContr := {PauliMatrix.asConsTensor |  $\nu \alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCo := { $\eta'$  |  $\mu \nu \otimes \text{pauliContr}$  |  $\nu \alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCoDown := {pauliCo |  $\mu \alpha \beta \otimes \epsilon_L$  |  $\alpha \alpha' \otimes \epsilon_R$  |  $\beta \beta'$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContrDown := {pauliContr |  $\mu \alpha \beta \otimes \epsilon_L$  |  $\alpha \alpha' \otimes \epsilon_R$  |  $\beta \beta'$ }T.tensor

```

The first of these definitions depends on `PauliMatrix.asConsTensor` which is defined as using an explicit basis expansion.

In these expressions we have the appearance of metrics  $\eta'$ , is the metric usually written as  $\eta_{\mu\nu}$ , `js:` etc.

With these we can define bispinors

```

/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p^{\mu}$ '. -/
def contrBispinorUp (p : complexContr) :=
  {pauliCo |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p^{\mu}$ '. -/
def contrBispinorDown (p : complexContr) :=
  { $\epsilon_L$  |  $\alpha \alpha' \otimes \epsilon_R$  |  $\beta \beta' \otimes \text{contrBispinorUp } p$  |  $\alpha \beta$ }T.tensor

/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p_{\mu}$ '. -/
def coBispinorUp (p : complexCo) := {pauliContr |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p_{\mu}$ '. -/
def coBispinorDown (p : complexCo) :=
  { $\epsilon_L$  |  $\alpha \alpha' \otimes \epsilon_R$  |  $\beta \beta' \otimes \text{coBispinorUp } p$  |  $\alpha \beta$ }T.tensor

```

Here `complexContr` and `complexCo` are complex contravariant and covariant Lorentz vectors. Lean knows to treat these as tensors when they appear in tensor expressions.

Using these definitions we can start to prove results about the pauli matrices and bispinors. These proofs rely on essentially the sorts of manipulations in the last section, although in some cases we expand tensors in terms of a basis and use rules about how the basis interacts with the operations in a tensor tree.

Examples of things we have proven range

```

lemma coBispinorDown_eq_pauliContrDown_contr (p : complexCo) :
  {coBispinorDown p |  $\alpha \beta = \text{pauliContrDown}$  |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T := by

```

the proof of which is an application of associativity of the tensor product, and appropriately shuffling around of the contractions.

To more complicated results such as

```

/-- The statement that ' $\eta_{\{\mu\nu\}} \sigma^{\{\mu \alpha \text{ dot } \beta\}} \sigma^{\{v \alpha' \text{ dot } \beta'\}} = 2 \epsilon^{\{\alpha\alpha'\}} \epsilon^{\{\text{dot } \beta \text{ dot } \beta'\}}$ '. -/
theorem pauliCo_contr_pauliContr :
  {pauliCo |  $v \alpha \beta \otimes \text{pauliContr} \mid v \alpha' \beta' = 2 \epsilon_L \mid \alpha \alpha' \otimes \epsilon_R \mid \beta \beta'\}^T := by$ 
```

#### 4. FUTURE WORK

The scale of formalizing all results regarding index notation is a task that surpasses the capacity of any single individual. Inspired by the Lean community’s blueprint projects, we have added to HepLean informal lemmas related to index notation and tensors. An example of such is

```

informal_lemma coBispinorUp_eq_metric_contr_coBispinorDown where
  math :≈ "{coBispinorUp p |  $\alpha \beta = \epsilon_L \mid \alpha \alpha' \otimes \epsilon_R \mid \beta \beta' \otimes \text{coBispinorDown p} \mid \alpha', \beta'\}^T"$ 
  proof :≈ "Expand ‘coBispinorDown’ and use fact that metrics contract to the identity."
  deps :≈ ["coBispinorUp", "coBispinorDown", "leftMetric", "rightMetric"]

```

The lemmas resource that we hope others—or even automated systems—will formalize in the future.

As demonstrated in our earlier examples, manipulating tensor expressions can involve tedious calculations, especially when dealing with directly with tensor trees. In the future, we intend to automate many of these routine steps by developing suitable tactics within Lean. We are optimistic that the structured nature of tensor trees will lend itself well to such automation, thereby streamlining computations and enhancing the efficiency of formal proofs involving index notation and tensor species.

There are two primary directions in which we can extend the concepts presented in this work. First, we could incorporate the spinor-helicity formalism, which is used in the study of scattering amplitudes. Second, we could extend our approach to encompass tensor *fields*, their derivatives etc. We do not anticipate any insurmountable challenges in pursuing these extensions. They represent promising avenues for future research and have the potential to significantly enhance the utility of formal methods in physics.

#### REFERENCES